

Funzioni per verifica e progetto a flessione composta

' A.G. 11/12/05

Public Function MRd(B, h, c, Asx, afcd, fyd, NRd)

```
Dim Ncmax As Double, niM As Double, Mcmax As Double
Dim Nsmax As Double, Msmax As Double
Ncmax = B * h * afcd / 10
niM = 289 / 594
Mcmax = 289 / 2376 * B * h ^ 2 * afcd / 1000
Nsmax = 2 * Asx * fyd / 10
Msmax = Asx * (h - 2 * c) * fyd / 1000
If NRd > Nsmax Then
    MRd = 0
ElseIf NRd > 0 Then
    MRd = Msmax * (1 - NRd / Nsmax)
ElseIf NRd > -niM * Ncmax Then
    MRd = Mcmax * (1 - ((NRd + niM * Ncmax) / (niM * Ncmax)) ^ 2) + Msmax
ElseIf NRd > -(Ncmax + Nsmax) Then
    n = 1 + (niM * Ncmax / ((1 - niM) * Ncmax + Nsmax)) ^ 2
    MRd = (Mcmax + Msmax) * (1 - (Abs(NRd + niM * Ncmax) / ((1 - niM) * Ncmax + Nsmax)) ^ n)
Else
    MRd = 0
End If
```

End Function

Public Function Asnec(b, h, c, afcd, fyd, MSd, NSd)

```
Dim Ncmax As Double, niM As Double, Mcmax As Double
Dim MsSd As Double, As_nec As Double, As_necP As Double
Ncmax = b * h * afcd / 10
niM = 289 / 594
Mcmax = 289 / 2376 * b * h ^ 2 * afcd / 1000
MsSd = Abs(MSd) - Mcmax * (1 - ((NSd + niM * Ncmax) / (niM * Ncmax)) ^ 2)
As_nec = Max(0, MsSd / ((h - 2 * c) * fyd) * 1000)
' itera per migliorare l'approssimazione del risultato
Do
    As_necP = As_nec
    MsSd = Abs(MSd) - MRd(b, h, c, As_necP, afcd, fyd, NSd)
    As_nec = Max(0, As_necP + MsSd / ((h - 2 * c) * fyd) * 1000)
Loop Until Abs(As_nec - As_necP) < 0.1
Asnec = As_nec
```

End Function

Public Function MxRd(B, h, c, Asx, afcd, fyd, NRd)

```
MxRd = MRd(B, h, c, Asx, afcd, fyd, NRd)
```

End Function

Public Function MyRd(B, h, c, Asy, afcd, fyd, NRd)

```
MyRd = MRd(h, B, c, Asy, afcd, fyd, NRd)
```

End Function

Public Function VerificaMN(B, h, c, Asx, Asy, afcd, fyd, MxSd, MySd, NSd)

```

If MxSd <> 0 Or MySd <> 0 Then
    ' --- per flessione composta
    Dim Mx_Rd As Double, My_Rd As Double
    If MxSd = 0 Then
        Mx_Rd = 1 ' valore qualsiasi, tanto il rapporto MSd/MRd è zero
    Else
        Mx_Rd = MRd(B, h, c, Asx, afcd, fyd, NSd)
    End If
    If MySd = 0 Then
        My_Rd = 1 ' valore qualsiasi, tanto il rapporto MSd/MRd è zero
    Else
        My_Rd = MRd(h, B, c, Asy, afcd, fyd, NSd)
    End If
    If Mx_Rd = 0 Or My_Rd = 0 Then
        ' possono essere nulli solo se contemporaneamente MSd<>0
        VerificaMN = 999.999
    Else
        VerificaMN = Abs(MxSd / Mx_Rd) ^ 1.5 + Abs(MySd / My_Rd) ^ 1.5
    End If
Else
    ' --- in presenza di solo sforzo normale
    Dim Nsmax As Double, Ncmax As Double
    If NSd = 0 Then
        Nsmax = 1 ' valore qualsiasi, tanto il rapporto NSd/NRd è zero
    Else
        ' nota: poiché alcune barre potrebbero appartenere sia ad Asx che Asy
        ' considero solo una delle due armature (la massima)
        Nsmax = 2 * Max(Asx, Asy) * fyd / 10
    End If
    Ncmax = B * h * afcd / 10
    If NSd > 0 Then
        ' trazione
        If Nsmax = 0 Then
            ' può essere nullo solo se contemporaneamente NSd<>0
            VerificaMN = 999.999
        Else
            VerificaMN = NSd / Nsmax
        End If
    Else
        ' compressione
        VerificaMN = Abs(NSd) / (Ncmax + Nsmax)
    End If
End If
End Function

Private Function Max(A, B)

    If A > B Then Max = A Else Max = B

End Function

```